

Realizar un Respaldo Programado con SQL Server 2005.

Dirigido a: Distribuidores Implementadores.

Por: Carlos Rosas.

1. Script para realizar copias de seguridad de su base de datos.

Es necesario realizar un script o comando SQL para realizar una copia de seguridad programada.

Se debe especificar la base de datos de la que queremos hacer nuestro respaldo, la ruta de almacenamiento del mismo y el nombre del archivo de respaldo, así como las características que deseemos configurar para realizar nuestra tarea.

En concreto, para seguir este ejemplo, el script que hemos creado es el siguiente:

```
-- Ruta para los respaldos
DECLARE @path VARCHAR(256)
SET @path = 'C:\Respaldos\'

DECLARE @DBName varchar(255)
DECLARE @DATABASES_Fetch int

DECLARE DATABASES_CURSOR CURSOR FOR
select DATABASE_NAME = db_name(s_mf.database_id)
from sys.master_files s_mf
where s_mf.state = 0
and has_dbaccess(db_name(s_mf.database_id)) = 1

-- Excluir bases de datos de SQL Server
and db_name(s_mf.database_id) not in ('Master', 'tempdb', 'model', 'msdb')
group by s_mf.database_id
order by 1

OPEN DATABASES_CURSOR
FETCH NEXT FROM DATABASES_CURSOR INTO @DBName

WHILE @@FETCH_STATUS = 0 BEGIN
declare @DBFileName varchar(256)
set @DBFileName = CAST(DATEPART(yyyy, getdate()) AS VARCHAR(4)) + '-' +
CAST(DATEPART(mm, getdate()) AS VARCHAR(2)) + '-' + CAST(DATEPART(dd, getdate()) AS
VARCHAR(2)) + '-' +
replace(replace(@DBName, ':', '_'), '\', '_') + '.bak'

PRINT '-- Realizando copia de seguridad ' + @path + @DBFileName
exec ('BACKUP DATABASE [' + @DBName + '] TO DISK = N''' + @path + @DBFileName + ''')

IF(SELECT recovery_model_desc FROM sys.databases WHERE name = @DBName) != 'SIMPLE' BEGIN
set @DBFileName = CAST(DATEPART(yyyy, getdate()) AS VARCHAR(4)) + '-' +
CAST(DATEPART(mm, getdate()) AS VARCHAR(2)) + '-' + CAST(DATEPART(dd, getdate()) AS
VARCHAR(2)) + '-' +
replace(replace(@DBName, ':', '_'), '\', '_') + '-LOG.bak'

PRINT '-- Realizando LOG Backup ' + @path + @DBFileName
exec ('BACKUP LOG [' + @DBName + '] TO DISK = N''' + @path + @DBFileName + ''')
END

FETCH NEXT FROM DATABASES_CURSOR INTO @DBName
END

CLOSE DATABASES_CURSOR
DEALLOCATE DATABASES_CURSOR
GO
```

El script anterior realizará la copia de seguridad de todas las bases de datos que se encuentren en línea en ese momento en el servidor de SQL Server y a las cuales el usuario que ejecute el script tenga acceso.

2. Si la versión de SQL instalada no es Express, puede utilizar un job del agente de SQL para programar la ejecución periódica de este script.

SQL Server Express, no contiene el componente SQL Agent, por lo tanto no es posible programar tareas, sin embargo, si se instala el componente Service Broker es posible mediante scripts obtener una funcionalidad equivalente:

- Elegir la base de datos que se va a respaldar.

```
Use Prueba
go
```

```
-- Crear una tabla en la base de datos,
-- con el propósito de mantener un log de respaldos
```

```
CREATE TABLE BackStatus
( BackUpId int identity primary key,
  DatabaseName sysname,
  LastBackupTime datetime,
  LastBackupFile nvarchar(200),
  BackupType nvarchar(20),
  BackupStatus nvarchar(1000) )
```

```
-- Crear a Service Broker
CREATE QUEUE [BackUpQueue]
CREATE SERVICE [BackUpService] ON QUEUE [BackUpQueue] ([DEFAULT]) ;
CREATE SERVICE [BackUpResponseService] ON QUEUE [BackUpQueue];
GO
```

- Copiar el código anterior y ejecutarlo en Management Studio Express.

```
-- Crear un procedimiento almacenado que administra el timer de respaldos
```

```
ALTER PROCEDURE [dbo].[HandleBackup] AS
DECLARE @BackupFileName nvarchar(200)
DECLARE @BackupInterval int
DECLARE @BackupStatus nvarchar(1000)
DECLARE @DBName sysname
DECLARE @conversationHandle uniqueidentifier
DECLARE @message_type_name sysname
DECLARE @dialog uniqueidentifier
-- Parámetros de la copia de seguridad
```

```
SET @DBName = 'Prueba'
SET @BackupFileName = 'C:\Respaldos\' + @DBName +
convert(nvarchar(20), getdate(), 112) + '_' +
replace(convert(nvarchar(20), GETDATE(), 108), ':', '-') + N'.bak'
```

```
-- Establecer el intervalo de respaldos 24 horas = 86400 segundos
```

```
SET @BackupInterval = 86400
```

```
BEGIN TRANSACTION
WAITFOR (RECEIVE top(1) @message_type_name=message_type_name,
```

```

        @dialog = conversation_handle
    FROM [BackUpQueue]), TIMEOUT 500

    IF (@message_type_name =
'http://schemas.microsoft.com/SQL/ServiceBroker/DialogTimer') BEGIN
        BEGIN CONVERSATION TIMER ( @dialog )
            TIMEOUT = @BackupInterval;
            COMMIT TRANSACTION
        BEGIN TRY
            BACKUP DATABASE @DBName TO DISK = @BackupFileName
                WITH CHECKSUM, INIT;
        END TRY
        BEGIN CATCH
            SET @BackupStatus = ERROR_MESSAGE()
        END CATCH;

        INSERT INTO BackStatus ( DatabaseName,
LastBackupTime, LastBackupFile, BackupType, BackupStatus)
VALUES (@DBName, GETDATE(), @BackupFileName, 'Database',
@BackupStatus);
    END ELSE
    IF (@message_type_name =
'http://schemas.microsoft.com/SQL/ServiceBroker/EndDialog')
    BEGIN
        END CONVERSATION @dialog
        COMMIT TRANSACTION
    END ELSE IF (@message_type_name =
'http://schemas.microsoft.com/SQL/ServiceBroker/Error')
    BEGIN
        END CONVERSATION @dialog
        COMMIT TRANSACTION
    END
END
go

ALTER QUEUE [BackUpQueue] WITH ACTIVATION (
    STATUS = ON, PROCEDURE_NAME = HandleBackUp ,
    MAX_QUEUE_READERS = 1, EXECUTE AS SELF)
Go

```

- Copiar el código anterior y ejecutarlo en Management Studio Express.

```

-- Estas sentencias solo se ejecutan una vez
-- para iniciar el proceso process running.
-- Una vez iniciado el proceso es permanente

```

```

DECLARE @conversationHandle uniqueidentifier
DECLARE @Timeout int

```

```

-- Esto configura el respaldo para iniciar a la medianoche diariamente
-- Si es necesario hacerlo de otra manera, calcule en numero de segundos
-- entre este momento y el tiempo en que se desee el primer backup

```

```

SET @Timeout = DATEDIFF(SECOND, GETDATE(),
    DATEADD (DAY, 1, convert(nvarchar(20), GETDATE(), 112)));
BEGIN DIALOG CONVERSATION @conversationHandle
FROM SERVICE [BackUpResponseService]
TO SERVICE 'BackUpService';
BEGIN CONVERSATION TIMER ( @ConversationHandle )
TIMEOUT = @Timeout;

```

E-mail: Desarrollo@MyBusinessPOS.com